

Dipolo frente a cilindro con circulación atrapada

Pablo Cobelli

```
In[988]:= ClearAll["Global`*"];  
Print["Esta celda borra todas las definiciones de variables  
previamente establecidas."]
```


Esta celda borra todas las definiciones de variables previamente establecidas.

Enunciado del problema

Tenemos un cilindro infinito de radio a y con circulación atrapada Γ ; inmerso en un fluido incompresible de densidad ρ que fluye de forma irrotacional. A una distancia $2a$ se encuentra un dipolo de intensidad μ orientado como indica la figura. Se pide hallar el valor de μ para que la fuerza sobre el cilindro sea nula.

```
In[989]:= Graphics[{GrayLevel[0.8], Disk[{0, 0}, 0.5], Red,  
Arrowheads[0.025], Arrow[{{1, -0.05}, {1, 0.05}}], Dashed,  
Black, Arrowheads[{-0.025, 0.025}], Arrow[{{0.5, 0}, {1, 0}}],  
Text["a", {0.75, 0.05}], Arrow[{{0, 0}, 1/4 * {Sqrt[2], Sqrt[2]}]},  
Text["a", 1/8 * {Sqrt[2], Sqrt[2]} + {-1, 1}/20], Text[" $\mu$ ", {1.05, 0}],  
Circle[{0, 0}, 0.6, Pi/2 * {0.55, 1.45}], Arrowheads[0.05],  
Arrow[{{0.01, 0.6}, {-0.01, 0.6}}], Text[" $\Gamma$ ", {0, 0.7}]}]
```

Out[989]=



Determinación del potencial complejo completo

Antes de comenzar vamos a *declarar* que determinadas variables serán reales (algunas, además, positivas) y otras, complejas. Para ello, hacemos la declaración:

In[990]:= **\$Assumptions =**
 $\{\mu \in \text{Reals}, z \in \text{Complexes}, a \in \text{Reals}, a > 0, \Gamma \in \text{Reals}, \Gamma > 0, \rho \in \text{Reals}, \rho > 0\}$
 Out[990]= $\{\mu \in \mathbb{R}, z \in \mathbb{C}, a \in \mathbb{R}, a > 0, \Gamma \in \mathbb{R}, \Gamma > 0, \rho \in \mathbb{R}, \rho > 0\}$

Potencial complejo del dipolo aislado

El potencial complejo inicial $W_{\text{dip}}(z)$ sera el correspondiente al dipolo exclusivamente. En general un dipolo de momento dipolar $\mu > 0$ ubicado en el punto z_0 , cuyo eje forma un angulo α con el eje real viene dado por un potencial complejo de la forma

In[991]:= **HoldForm**[$W_{\text{dipolo generico}}[z] = \mu \text{Exp}[i \alpha] / (2 * \text{Pi} * (z - z_0))$]

Out[991]= $W_{\text{dipolo generico}}(z) = \frac{\mu \exp(i \alpha)}{2 \pi (z - z_0)}$

Segun la figura, nuestro dipolo forma un angulo de $\pi/2$ radianes con el eje real (x), y esta ubicado en el punto $z=2a$, con lo cual tenemos:

In[992]:= **HoldForm**[$W_{\text{dipolo particular}}[z] = i \mu / (2 * \text{Pi} * (z - z_0))$]

Out[992]= $W_{\text{dipolo particular}}(z) = \frac{i \mu}{2 \pi (z - z_0)}$

Definimos esto como nuestra funcion "Wdip" para poder trabajar mas comodis en *Mathematica*.

In[993]:= **Wdip**[z_] := $i * \mu / 2 / \text{Pi} * (1 / (z - 2 * a))$

Inclusion del cilindro (contorno solido) via el Teorema del Circulo

Ahora tenemos que agregar el contorno del cilindro haciendo que su frontera sea una linea de corriente del flujo. Esto se logra aplicando el teorema del circulo, que podemos aplicar facilmente haciendo:

In[994]:= **Wdipycil**[z_] := **Wdip**[z] + **Conjugate**[**Wdip**[a^2 / **Conjugate**[z]]]

In[995]:= **Refine**[**Wdipycil**[z]]

Out[995]= $\frac{i \mu}{2 \pi (z - 2 a)} - \frac{i \mu}{2 \pi \left(\frac{a^2}{z} - 2 a\right)}$

Y ahora la funcion "Wdipycil" describe el potencial complejo del dipolo y el cilindro.

Observemos que, al aplicar el teorema del circulo, la 'imagen' del dipolo se ubica naturalmente en $z=a/2$ (sobre el eje x).

Agregado de la circulacion 'atrapada' en torno al cilindro

Ahora agregamos la circulacion atrapada, recordando que un vortice de circulacion Γ tiene potencial complejo de la forma:

Recordemos que un vortice de circulacion $\Gamma > 0$ ubicado en z_0 tiene potencial complejo de la forma

In[996]:= **HoldForm**[$W_{\text{vortice generico}}[z] = \Gamma / (2 * \text{Pi} * i) * \text{Log}[z - z_0]$]

Out[996]= $W_{\text{vortice generico}}(z) = \frac{\Gamma \log(z - z_0)}{2 \pi i}$

Este vortice tiene una circulacion (positiva) en el sentido matematico, es decir, en el sentido opuesto al de las agujas del reloj (i.e., el mismo que se muestra en la figura del enunciado).

Definimos entonces la funcion "Wvor" como el potencial complejo asociado a la circulacion atrapada. Hacemos

```
In[997]:= Wvor[z_] := -i * Γ / 2 / Pi * Log[z]
```

Y finalmente definimos el potencial complejo total del problema como la suma del correspondiente al dipolo+cilindro y el que acabamos de escribir para el vortice en el origen:

```
In[998]:= W[z_] := Wdipycil[z] + Wvor[z]
```

y obtenemos una expresion mas 'limpia' para el potencial complejo:

```
In[999]:= Refine[W[z]]
```

$$\text{Out[999]} = -\frac{i \mu}{2 \pi \left(\frac{a^2}{z} - 2 a\right)} + \frac{i \mu}{2 \pi (z - 2 a)} - \frac{i \Gamma \log(z)}{2 \pi}$$

Calculamos entonces la derivada del potencial complejo (es decir, estamos calculando la velocidad conjugada u^*),

```
In[1000]:= Refine[D[W[z], z]]
```

$$\text{Out[1000]} = -\frac{i a^2 \mu}{2 \pi z^2 \left(\frac{a^2}{z} - 2 a\right)^2} - \frac{i \mu}{2 \pi (z - 2 a)^2} - \frac{i \Gamma}{2 \pi z}$$

y su cuadrado, que nos sera util para el calculo de la fuerza via el Teorema de Blasius:

```
In[1001]:= Refine[D[W[z], z]] ^ 2
```

$$\text{Out[1001]} = \left(-\frac{i a^2 \mu}{2 \pi z^2 \left(\frac{a^2}{z} - 2 a\right)^2} - \frac{i \mu}{2 \pi (z - 2 a)^2} - \frac{i \Gamma}{2 \pi z} \right)^2$$

Definimos entonces la variable "dWdz2" para que coincida con el valor de $\left(\frac{dW}{dz}\right)^2$:

```
In[1002]:= dWdz2 = Refine[D[W[z], z]] ^ 2
```

$$\text{Out[1002]} = \left(-\frac{i a^2 \mu}{2 \pi z^2 \left(\frac{a^2}{z} - 2 a\right)^2} - \frac{i \mu}{2 \pi (z - 2 a)^2} - \frac{i \Gamma}{2 \pi z} \right)^2$$

Si expandimos el cuadrado, encontramos los siguientes terminos, que 'guardamos' en la variable 'todos' para analizar su contribucion, termino a termino, en la proxima seccion.

```
In[1003]:= todos = Expand[dWdz2]
```

$$\text{Out[1003]} = \frac{a^2 \Gamma \mu}{2 \pi^2 z^3 \left(\frac{a^2}{z} - 2 a\right)^2} - \frac{a^2 \mu^2}{2 \pi^2 z^2 \left(\frac{a^2}{z} - 2 a\right)^2 (z - 2 a)^2} - \frac{a^4 \mu^2}{4 \pi^2 z^4 \left(\frac{a^2}{z} - 2 a\right)^4} - \frac{\Gamma \mu}{2 \pi^2 z (z - 2 a)^2} - \frac{\mu^2}{4 \pi^2 (z - 2 a)^4} - \frac{\Gamma^2}{4 \pi^2 z^2}$$

Analisis termino a termino

Ahora tomamos cada sumando por separado, para analizar termino a termino las contribuciones a los residuos de cada uno de ellos. Recordemos que se tratara de los residuos internos al contorno (frontera del solido), es decir en $z=0$ (vortice) y $z=a/2$ (dipolo imagen introducido mediante el teorema del circulo).

Creamos entonces una lista de los sumandos (con su signo) que aparecen en el potencial complejo.

In[1004]= **Table[T[i] = Part[todos, i], {i, 1, 6}]**

$$\text{Out[1004]= } \left\{ -\frac{\Gamma^2}{4\pi^2 z^2}, -\frac{a^2 \Gamma \mu}{2\pi^2 z^3 \left(\frac{a^2}{z} - 2a\right)^2}, -\frac{\Gamma \mu}{2\pi^2 z(z-2a)^2}, \right. \\ \left. -\frac{a^4 \mu^2}{4\pi^2 z^4 \left(\frac{a^2}{z} - 2a\right)^4}, -\frac{\mu^2}{4\pi^2 (z-2a)^4}, -\frac{a^2 \mu^2}{2\pi^2 z^2 \left(\frac{a^2}{z} - 2a\right)^2 (z-2a)^2} \right\}$$

Arrancamos entonces con el analisis; vamos a calcular los residuos en $z=0$ y $z=a/2$ (i.e., internos al contorno) para cada uno de los sumandos de $\left(\frac{dW}{dz}\right)^2$ que calculamos al final de la seccion precedente.

Este primer termino corresponde al cuadrado de la singularidad en $z=0$ (int-int):

In[1005]= **Residue[T[1], {z, 0}] + Residue[T[1], {z, a/2}]**

Out[1005]= 0

Observemos que sucede si, en lugar de considerar la suma de ambos residuos, calculamos cada uno por separado:

In[1006]= **Residue[T[1], {z, 0}]**
Residue[T[1], {z, a/2}]

Out[1006]= 0

Out[1007]= 0

Este segundo corresponde al producto de las sings. en $z = 0$ y $z = a/2$ (int - int):

In[1008]= **Residue[T[2], {z, 0}] + Residue[T[2], {z, a/2}]**

Out[1008]= 0

Observemos que sucede si, en lugar de considerar la suma de ambos residuos, calculamos cada uno por separado:

In[1009]= **Residue[T[2], {z, 0}]**
Residue[T[2], {z, a/2}]

$$\text{Out[1009]= } -\frac{\Gamma \mu}{2\pi^2 a^2}$$

$$\text{Out[1010]= } \frac{\Gamma \mu}{2\pi^2 a^2}$$

Este tercer termino corresponde al producto de las sings. en $z=0$ y $z=2a$ (int-ext):

In[1011]= **Residue[T[3], {z, 0}] + Residue[T[3], {z, a/2}]**

$$\text{Out[1011]= } -\frac{\Gamma \mu}{8\pi^2 a^2}$$

Observemos que sucede si, en lugar de considerar la suma de ambos residuos, calculamos cada uno por separado:

In[1012]:= **Residue**[**T**[3], {z, 0}]
Residue[**T**[3], {z, a / 2}]

$$\text{Out[1012]} = -\frac{\Gamma \mu}{8 \pi^2 a^2}$$

Out[1013]= 0

Este cuarto termino corresponde al cuadrado de la singularidad en $z=a/2$ (int-int):

In[1014]:= **Residue**[**T**[4], {z, 0}] + **Residue**[**T**[4], {z, a / 2}]

Out[1014]= 0

Repetimos el calculo de cada residuo por separado:

In[1015]:= **Residue**[**T**[4], {z, 0}]
Residue[**T**[4], {z, a / 2}]

Out[1015]= 0

Out[1016]= 0

El quinto termino corresponde al cuadrado de la singularidad en $z=2a$ (ext-ext):

In[1017]:= **Residue**[**T**[5], {z, 0}] + **Residue**[**T**[5], {z, a / 2}]

Out[1017]= 0

Repetimos tambien el calculo de cada residuo por separado:

In[1018]:= **Residue**[**T**[5], {z, 0}]
Residue[**T**[5], {z, a / 2}]

Out[1018]= 0

Out[1019]= 0

El ultimo termino (sexto) corresponde al producto de las sings. en $z=a/2$ y $z=2a$ (int-ext):

In[1020]:= **Residue**[**T**[6], {z, 0}] + **Residue**[**T**[6], {z, a / 2}]

$$\text{Out[1020]} = -\frac{2 \mu^2}{27 \pi^2 a^3}$$

Por completitud, cada residuo por separado da:

In[1021]:= **Residue**[**T**[6], {z, 0}]
Residue[**T**[6], {z, a / 2}]

Out[1021]= 0

$$\text{Out[1022]} = -\frac{2 \mu^2}{27 \pi^2 a^3}$$

Segun vemos, solo contribuyen a los residuos los terminos que cruzan singularidades entre el exterior y el interior del contorno cerrado que es frontera del solido (cilindro). Y los residuos solo los calculamos en las singularidades internas (es una hipotesis del teorema de residuos) ubicadas en $z=0$ (correspondiente al vortice que da la circulacion atrapada en el cilindro) y $z=a/2$ (el dipolo imagen dentro del cilindro).

Entonces, si calculamos los residuos de todos los terminos en $z=0$ y $z=a/2$ (es decir, la suma de todos los anteriores), resulta:

In[1023]= **Residue**[dWdz2, {z, 0}] + **Residue**[dWdz2, {z, a / 2}]

$$\text{Out[1023]= } \frac{27 a \Gamma \mu - 4 \mu^2}{54 \pi^2 a^3} - \frac{5 \Gamma \mu}{8 \pi^2 a^2}$$

Observemos que da lo mismo que sumar las contribuciones de cada termino por separado:

In[1024]= **Residue**[**T**[1] + **T**[2] + **T**[3] + **T**[4] + **T**[5] + **T**[6], {z, 0}] +
Residue[**T**[1] + **T**[2] + **T**[3] + **T**[4] + **T**[5] + **T**[6], {z, a / 2}]

$$\text{Out[1024]= } \frac{27 a \Gamma \mu - 4 \mu^2}{54 \pi^2 a^3} - \frac{5 \Gamma \mu}{8 \pi^2 a^2}$$

Y finalmente simplificamos un poco la expresion obtenida para poder reconocer los terminos que la integran:

In[1025]= **Apart** [%]

$$\text{Out[1025]= } -\frac{2 \mu^2}{27 \pi^2 a^3} - \frac{\Gamma \mu}{8 \pi^2 a^2}$$

Siguiendo con la idea que les comente en clase...

Segun les comente en clase, **no** es necesario calcular los residuos en $z=0$ y $z=a/2$ **para todos los 6 terminos de $\left(\frac{dW}{dz}\right)^2$.**

En clase vimos que podemos tomar exclusivamente **2** de esos sumandos: (llamando "dWdz2clase" a la variable que representa al termino que escribimos en clase)

In[1026]= **dWdz2clase** = **2 * (-i / 2 / Pi) ^ 2 ***
(Gamma / z * mu / (z - 2 * a) ^ 2 + mu ^ 2 * a ^ 2 / (z - 2 * a) ^ 2 / (a ^ 2 - 2 * a * z) ^ 2)

$$\text{Out[1026]= } -\frac{\frac{a^2 \mu^2}{(z-2a)^2 (a^2-2az)^2} + \frac{\Gamma \mu}{z(z-2a)^2}}{2 \pi^2}$$

donde el factor 2 al inicio de la ultima expresion representa, como dijimos, la contribucion de los dobles productos del cuadrado del potencial complejo. Observen que los terminos que retuvimos de $\left(\frac{dW}{dz}\right)^2$ son los correspondientes al producto de: (1) $z=0$ y $z=2a$, y (2) $z=a/2$ y $z=2a$.

El calculo de residuos sobre esta expresion nos dara:

In[1027]= **SumaDeResiduos** = **Residue**[dWdz2clase, {z, 0}] + **Residue**[dWdz2clase, {z, a / 2}]

$$\text{Out[1027]= } -\frac{2 \mu^2}{27 \pi^2 a^3} - \frac{\Gamma \mu}{8 \pi^2 a^2}$$

resultado que coincide con el obtenido al final de la seccion anterior.

Como calcular los residuos **sin Mathematica**

Mathematica calcula los residuos de una funcion en un punto en forma directa; recordemos como calcularlos 'a mano'. Para ello consideremos, como ejemplo, uno de los sumandos del potencial complejo, el que hemos llamado T(3) y que corresponde a

In[1028]:= **T[3]**

$$\text{Out[1028]} = -\frac{\Gamma \mu}{2 \pi^2 z (z - 2 a)^2}$$

Vamos a calcular sus residuos en las singularidades interiores al contorno (cilindro de radio a), es decir, en $z=0$ y $z=a/2$. Para ello comenzamos separando en fracciones simples respecto de la variable z :

In[1029]:= **DescompFraccionesSimples = Apart[T[3], z]**

$$\text{Out[1029]} = -\frac{\Gamma \mu}{8 \pi^2 a^2 z} + \frac{\Gamma \mu}{8 \pi^2 a^2 (z - 2 a)} - \frac{\Gamma \mu}{4 \pi^2 a (z - 2 a)^2}$$

Vemos que en $z=a/2$ este termino no presenta singularidad, por lo tanto tampoco residuo.

En $z = 0$, por el contrario, existe una singularidad y su residuo es sencillo de calcular, ya que corresponde al factor multiplicativo de $(z - z_0)^{-1} = (z - 0)^{-1}$ y resulta:

In[1030]:= **Coefficient[DescompFraccionesSimples, z, -1]**

$$\text{Out[1030]} = -\frac{\Gamma \mu}{8 \pi^2 a^2}$$

cuyo valor coincide con el que calculamos directamente antes.

Analicemos, como ultimo ejemplo de este calculo, el termino T(6), que era de la forma

In[1031]:= **T[6]**

$$\text{Out[1031]} = -\frac{a^2 \mu^2}{2 \pi^2 z^2 \left(\frac{a}{z} - 2 a\right)^2 (z - 2 a)^2}$$

Su descomposicion en fracciones simples da

In[1032]:= **DescompFraccionesSimples = Apart[T[6], z]**

$$\text{Out[1032]} = \frac{2 \mu^2}{27 \pi^2 a^3 (z - 2 a)} - \frac{4 \mu^2}{27 \pi^2 a^3 (2 z - a)} - \frac{\mu^2}{18 \pi^2 a^2 (z - 2 a)^2} - \frac{2 \mu^2}{9 \pi^2 a^2 (2 z - a)^2}$$

Reformulemos esta ultima expresion para que incluya terminos en $(z-a/2)$ y no en $(2*z-a)$. Esto no es imprescindible para el calculo, es conveniente para visualizar los terminos y asociarlos con las posiciones de las singularidades. Para ello hacemos:

In[1033]:= **DescompFraccionesSimples = DescompFraccionesSimples /. 2 * z - a → 2 * (z - a / 2)**

$$\text{Out[1033]} = \frac{2 \mu^2}{27 \pi^2 a^3 (z - 2 a)} - \frac{2 \mu^2}{27 \pi^2 a^3 \left(z - \frac{a}{2}\right)} - \frac{\mu^2}{18 \pi^2 a^2 (z - 2 a)^2} - \frac{\mu^2}{18 \pi^2 a^2 \left(z - \frac{a}{2}\right)^2}$$

Vemos que esta descomposicion pone en evidencia que tenemos singularidades en $z=a/2$ y en $z=2a$ (esta ultima no nos interesa porque es externa al contorno solido). En particular, el unico sumando que nos interesa es el segundo, ya que es el que es proporcional a $(z-a/2)^{-1}$. Su coeficiente es

In[1034]:= **Coefficient[DescompFraccionesSimples, (z - a / 2), -1]**

$$\text{Out[1034]} = -\frac{2 \mu^2}{27 \pi^2 a^3}$$

resultado que coincide con lo obtenido previamente usando directamente el comando 'Residue' de

Mathematica.

Para aquellos que quieran recordar como aplicar el metodo de descomposicion en fracciones parciales, les recomiendo (entre otros) la siguiente fuente:

Weisstein, Eric W. "Partial Fraction Decomposition." From MathWorld-- A Wolfram Web Resource. <http://mathworld.wolfram.com/PartialFractionDecomposition.html>

Calculo de la fuerza

Ahora que conocemos los valores de los residuos, solo queda calcular la fuerza que el flujo hace sobre el cilindro (por unidad de longitud del cilindro). Para ello, usamos el teorema de Blasius:

In[1035]= **HoldForm** [$F_x - i F_y == i \rho / 2 * \text{Integrate}[(dW/dz)^2, z]$]

Out[1035]=
$$F_x - i F_y = \frac{1}{2} i \rho \int \left(\frac{dW}{dz} \right)^2 dz$$

junto con el teorema de residuos,

In[1036]= **HoldForm** [$\text{Integrate}[f[z], z] == 2 \pi i \text{Sum}[\text{Res}[(dW/dz)^2; z_k], z_k]$]

Out[1036]=
$$\int f(z) dz = 2 \pi i \sum_{z_k} \text{Res} \left(\left(\frac{dW}{dz} \right)^2; z_k \right)$$

para obtener la relacion:

In[1037]= **HoldForm** [$F_x - i F_y == -\pi \rho * \text{Sum}[\text{Res}[(dW/dz)^2; z_k], z_k]$]

Out[1037]=
$$F_x - i F_y = -\pi \rho \sum_{z_k} \text{Res} \left(\left(\frac{dW}{dz} \right)^2; z_k \right)$$

En las igualdades anteriores, las integrales son integrales de camino cerrado sobre la frontera del obstaculo, y se sobreentiende que la sumatoria se realiza sobre todas las singularidades z_k de $\left(\frac{dW}{dz}\right)^2$ existentes dentro de dicho contorno.

Luego tenemos,

In[1038]= **FuerzaCompleja** = $-\pi * \rho * \text{SumaDeResiduos}$

Out[1038]=
$$-\pi \rho \left(-\frac{2 \mu^2}{27 \pi^2 a^3} - \frac{\Gamma \mu}{8 \pi^2 a^2} \right)$$

In[1039]= **Fx** = **ComplexExpand** [**Re** [**FuerzaCompleja**]]

Out[1039]=
$$\frac{2 \mu^2 \rho}{27 \pi a^3} + \frac{\Gamma \mu \rho}{8 \pi a^2}$$

In[1040]= **Fy** = **-ComplexExpand** [**Im** [**FuerzaCompleja**]]

Out[1040]= 0

En las ultimas dos lineas de calculo simbolico, el comando 'ComplexExpand' tiene el efecto de simplificar la expresion para que *Mathematica* calcule efectivamente las partes real e imaginarias de la fuerza.

Respuesta a la pregunta del ejercicio

El ejercicio requiere calcular el valor de μ para que la fuerza sobre el cilindro sea nula. Entonces hacemos:

```
In[1041]:= muSolucion = Solve[FuerzaCompleja == 0,  $\mu$ ]
```

```
Out[1041]:=  $\left\{ \left\{ \mu \rightarrow 0 \right\}, \left\{ \mu \rightarrow -\frac{27 a \Gamma}{16} \right\} \right\}$ 
```

Y observamos que tenemos dos soluciones, lo que es consecuencia de que la fuerza compleja que calculamos depende cuadráticamente de μ . La primera solución es trivial, corresponde a que no haya dipolo. La segunda es la más rica y es la que buscamos:

```
In[1042]:= muSolucion[[2]]
```

```
Out[1042]:=  $\left\{ \mu \rightarrow -\frac{27 a \Gamma}{16} \right\}$ 
```

Visualicemos gráficamente los resultados

Borramos cualquier valor precedente de los parámetros.

```
In[1043]:= ClearAll[a,  $\Gamma$ ,  $\mu$ ]
```

Construcción del potencial complejo

Graficamos primero las líneas de corriente, y de potencial de velocidades para el potencial debido al dipolo:


Damos un valor a ' μ ' y graficamos

```
In[1044]:= a = 1;  $\mu$  = 1;
```

```

In[1045]:= Show[ContourPlot[Im[Wdip[x + i * y]],
  {x, -6 a, 6 a}, {y, -6 a, 6 a}, AspectRatio -> Automatic,
  Contours -> Flatten[{Table[i, {i, -0.5, 1, 1 / 40}], Table[i, {i, 0, 1, 10}]}],
  PlotRange -> Automatic, PerformanceGoal -> "Quality", ContourStyle -> {Black},
  Axes -> True, AxesLabel -> {"x", "y"}, PlotLegends -> Automatic,
  ColorFunction -> "TemperatureMap", ContourPlot[Re[Wdip[x + i * y]],
  {x, -6 a, 6 a}, {y, -6 a, 6 a}, AspectRatio -> Automatic,
  Contours -> Flatten[{Table[i, {i, -0.5, 1, 1 / 40}], Table[i, {i, 0, 1, 10}]}],
  PlotRange -> All, PerformanceGoal -> "Quality",
  ContourShading -> None, ContourStyle -> {Dashed},
  Axes -> True, AxesLabel -> {"x", "y"}], ImageSize -> Large]

```




En este grafico vemos las lineas de corriente (en trazo continuo) y las lineas de potencial constante (en linea punteada). La posicion del dipolo con la flecha roja.

Agreguemos ahora el cilindro y veamos como se modificaron las lineas de corriente originales:

```

In[1046]:= a = 1;  $\mu = 1$ ;
Show[ContourPlot[Im[Wdipycil[x +  $i$  * y]],
  {x, -6 a, 6 a}, {y, -6 a, 6 a}, AspectRatio  $\rightarrow$  Automatic,
  Contours  $\rightarrow$  Flatten[{Table[i, {i, -0.5, 1, 1 / 40}], Table[i, {i, 0, 1, 10}]}],
  PlotRange  $\rightarrow$  Automatic, PerformanceGoal  $\rightarrow$  "Quality", ContourStyle  $\rightarrow$  {Black},
  Axes  $\rightarrow$  True, AxesLabel  $\rightarrow$  {"x", "y"}, PlotLegends  $\rightarrow$  Automatic,
  ColorFunction  $\rightarrow$  "TemperatureMap"], ContourPlot[Re[Wdipycil[x +  $i$  * y]],
  {x, -6 a, 6 a}, {y, -6 a, 6 a}, AspectRatio  $\rightarrow$  Automatic,
  Contours  $\rightarrow$  Flatten[{Table[i, {i, -0.5, 1, 1 / 40}], Table[i, {i, 0, 1, 10}]}],
  PlotRange  $\rightarrow$  All, PerformanceGoal  $\rightarrow$  "Quality", ContourShading  $\rightarrow$  None,
  ContourStyle  $\rightarrow$  {Dashed}, Axes  $\rightarrow$  True, AxesLabel  $\rightarrow$  {"x", "y"}],
Graphics[{Green, Circle[{0, 0}, a], Red, Arrowheads[0.02],
  Arrow[{2 a, -0.2}, {2 a, 0.2}]}], ImageSize  $\rightarrow$  Large]

```



En este grafico vemos las lineas de corriente (en trazo continuo) y las lineas de potencial constante (en linea punteada). La posicion del dipolo con la flecha roja, y la periferia del cilindro se denota con linea verde superpuesta. Vemos que efectivamente, la aplicacion del teorema del circulo (i.e., la inclusion de una singularidad tipo dipolo en $z=a/2$) hizo de la curva $|z|=a$ una linea de corriente.

En la seccion siguiente consideramos el potencial completo (dipolo + cilindro + circulacion atrapada) para dos sets de valores distintos de los parametros.

Graficamos para Γ y μ de valor unidad e igual signo

Comenzamos por borrar cualquier valor que a , Γ y μ tengan almacenado,

```
In[1048]:= ClearAll[a,  $\Gamma$ ,  $\mu$ ]
```

y luego le damos valores arbitrarios,


```
In[1049]:= a = 1;  $\Gamma$  = 1;  $\mu$  = 1;
```

Ahora graficamos las líneas de corriente y de potencial de velocidad:

```

In[1050]:= Show[ContourPlot[Im[W[x + i * y]],
  {x, -6 a, 6 a}, {y, -6 a, 6 a}, AspectRatio -> Automatic,
  Contours -> Flatten[{Table[i, {i, -0.5, 1, 1 / 40}], Table[i, {i, 0, 1, 10}]}],
  PlotRange -> Automatic, PerformanceGoal -> "Quality", ContourStyle -> {Black},
  Axes -> True, AxesLabel -> {"x", "y"}, PlotLegends -> Automatic,
  ColorFunction -> "TemperatureMap"], ContourPlot[Re[W[x + i * y]],
  {x, -6 a, 6 a}, {y, -6 a, 6 a}, AspectRatio -> Automatic,
  Contours -> Flatten[{Table[i, {i, -0.5, 1, 1 / 40}], Table[i, {i, 0, 1, 10}]}],
  PlotRange -> All, PerformanceGoal -> "Quality", ContourShading -> None,
  ContourStyle -> {Dashed}, Axes -> True, AxesLabel -> {"x", "y"}],
  Graphics[{GrayLevel[0.3], Disk[{0, 0}, a], Red, Arrowheads[0.02],
  Arrow[{2 a, -0.2}, {2 a, 0.2}]}], ImageSize -> Large]

```



En este grafico vemos las lineas de corriente (en trazo continuo) y las lineas de potencial constante (en linea punteada). La posicion del cilindro esta marcada con el disco gris y la posicion del dipolo con la flecha roja.

Finalmente, borramos los valores asignados a 'a', ' Γ ' y ' μ ' para evitar problemas posibles al reevaluar las celdas del notebook.

```
In[1051]:= ClearAll[a,  $\Gamma$ ,  $\mu$ ]
```

Graficamos para Γ y μ de valor unidad pero distinto signo

Comenzamos por borrar cualquier valor que a, Γ y μ tengan almacenado,

```
In[1052]:= ClearAll[a,  $\Gamma$ ,  $\mu$ ]
```

y luego le damos valores arbitrarios,


```
In[1053]:= a = 1;  $\Gamma$  = -1;  $\mu$  = 1;
```

Ahora graficamos las líneas de corriente y de potencial de velocidad:

```

In[1054]:= Show[ContourPlot[Im[W[x + i * y]],
  {x, -6 a, 6 a}, {y, -6 a, 6 a}, AspectRatio -> Automatic,
  Contours -> Flatten[{Table[i, {i, -0.5, 1, 1 / 40}], Table[i, {i, 0, 1, 10}]}],
  PlotRange -> Automatic, PerformanceGoal -> "Quality", ContourStyle -> {Black},
  Axes -> True, AxesLabel -> {"x", "y"}, PlotLegends -> Automatic,
  ColorFunction -> "TemperatureMap", ContourPlot[Re[W[x + i * y]],
  {x, -6 a, 6 a}, {y, -6 a, 6 a}, AspectRatio -> Automatic,
  Contours -> Flatten[{Table[i, {i, -0.5, 1, 1 / 40}], Table[i, {i, 0, 1, 10}]}],
  PlotRange -> All, PerformanceGoal -> "Quality", ContourShading -> None,
  ContourStyle -> {Dashed}, Axes -> True, AxesLabel -> {"x", "y"}],
  Graphics[{GrayLevel[0.3], Disk[{0, 0}, a], Red, Arrowheads[0.02],
  Arrow[{2 a, -0.2}, {2 a, 0.2}]}], ImageSize -> Large]

```



En este grafico vemos las lineas de corriente (en trazo continuo) y las lineas de potencial constante (en linea punteada). La posicion del cilindro esta marcada con el disco gris y la posicion del dipolo con la flecha roja.

Finalmente, borramos los valores asignados a 'a', ' Γ ' y ' μ ' para evitar problemas posibles al reevaluar las celdas del notebook.

```
In[1055]:= ClearAll[a,  $\Gamma$ ,  $\mu$ ]
```

Graficamos para el valor de μ para el cual no hay fuerza sobre el cilindro

Comenzamos por borrar cualquier valor que a , Γ y μ tengan almacenado,

```
In[1056]:= ClearAll[a,  $\Gamma$ ,  $\mu$ ]
```

y luego le damos los valores correspondientes al caso para el cual no habria fuerza sobre el cilindro,


```
In[1057]:= a = 1;  $\Gamma$  = 1;  $\mu$  = -27 / 16 * a *  $\Gamma$ ;
```

Ahora graficamos las lineas de corriente y de potencial de velocidad:


```

In[1058]:= Show[ {ContourPlot[Im[W[x + i * y]],
  {x, -6 a, 6 a}, {y, -6 a, 6 a}, AspectRatio -> Automatic,
  Contours -> Flatten[{Table[i, {i, -0.5, 1, 1 / 40}], Table[i, {i, 0, 1, 10}]}],
  PlotRange -> Automatic, PerformanceGoal -> "Quality", ContourStyle -> {Black},
  Axes -> True, AxesLabel -> {"x", "y"}, PlotLegends -> Automatic,
  ColorFunction -> "TemperatureMap"}, ContourPlot[Re[W[x + i * y]],
  {x, -6 a, 6 a}, {y, -6 a, 6 a}, AspectRatio -> Automatic,
  Contours -> Flatten[{Table[i, {i, -0.5, 1, 1 / 40}], Table[i, {i, 0, 1, 10}]}],
  PlotRange -> All, PerformanceGoal -> "Quality", ContourShading -> None,
  ContourStyle -> {Dashed}, Axes -> True, AxesLabel -> {"x", "y"}],
  Graphics[{GrayLevel[0.30], Disk[{0, 0}, a], Red, Arrowheads[0.02],
  Arrow[{{2 a, -0.2}, {2 a, 0.2}}]}], ImageSize -> Large]

```



En este grafico, al igual que en el precedente, se ilustran las lineas de corriente (en trazo continuo) y las lineas de potencial constante (en linea punteada). La posicion del cilindro esta marcada con el disco gris y la posicion del dipolo con la flecha roja.

Finalmente, borramos los valores asignados a 'a', ' Γ ' y ' μ ' para evitar problemas posibles al reevaluar las celdas del notebook.

In[1059]:= ClearAll[a, Γ , μ]

Como interpretar o que informacion obtener de los graficos de lineas de corriente

En primer lugar recordemos que estamos analizando una situacion que es estacionaria, por lo cual las lineas de corriente que calculamos son tambien las lineas de trayectoria de las particulas de fluido. Por otro lado, estamos despreciando la accion de cualquier fuerza externa sobre el fluido, tales como la fuerza gravitatoria, las debidas a friccion (por ser considerado ideal el flujo), etc. Esto significa que las unicas fuerzas actuantes sobre las particulas de fluido son debidas a los efectos de la presion (de los gradientes de presion, para ser exactos).

Consideremos entonces el movimiento de una particula de fluido sobre una linea de corriente curva. Asumamos, por simplicidad, que la velocidad de la particula sobre dicha linea es constante. (Esto implicaria, de acuerdo al teorema de Bernoulli, que la diferencia de presion sobre la linea de corriente es constante.) Ahora bien, al seguir la linea de corriente curva, la particula de fluido esta cambiando su direccion, con lo cual debe existir una fuerza centripeta actuando en forma normal (perpendicular) a la direccion del movimiento. Esta fuerza solo puede generarse por diferencias de presion (ya que, por hipotesis, no hay otras fuerzas actuando). A su vez, esto implica que la presion de un lado de la particula de fluido es mayor que en el lado opuesto. Como corolario, concluimos que **si una linea de corriente es curva, debe existir un gradiente de presion perpendicular a la linea, con la presion aumentando hacia afuera del centro de curvatura.**

A los que les interese, pueden consultar una derivacion rigurosa de esta relacion en el apendice de: Babinsky, H. "How do wings work?". Physics Education, Vol. 38, No. 6, pp. 497-503 (2003). disponible en el sitio web de la materia.

Con esta conclusion en mente, resulta facil interpretar los resultados graficos que obtuvimos para las lineas de corriente del problema en los diferentes casos que analizamos.